
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2016-06_D17wk10’.

2. Corrections required

2.1 Change 1 – Correction to function ‘f_EUTRA_NB_SubFrameTimingDuration’

	Function name
	f_EUTRA_NB_SubFrameTimingDuration ()

	Reason for change
	Incorrect time calculation with HSFN.

	Summary of change
	Incorrect time calculation with HSFN. 

	TTCN module
	EUTRA_NB_Timing.ttcn

	MCC160 Comment
	Accepted


Before change

  function f_EUTRA_NB_SubFrameTimingDuration(SubFrameTiming_Type p_T1,

                                             SubFrameTiming_Type p_T2) return integer

  {

    var integer v_MillisecondsT1;

    var integer v_MillisecondsT2;

    var integer v_SubframeNumberT1 := 0;

    var integer v_SubframeNumberT2 := 0;

    var integer v_MaxValue := 10240;   // SFN is 0..1023 in steps of 10ms

    var integer v_Duration;

    if (ischosen(p_T1.Subframe.Number) and

        ischosen(p_T2.Subframe.Number)) {

      v_SubframeNumberT1 := p_T1.Subframe.Number;

      v_SubframeNumberT2 := p_T2.Subframe.Number;

    }

    v_MillisecondsT1 := (p_T1.SFN.Number * 10) + v_SubframeNumberT1;
    v_MillisecondsT2 := (p_T2.SFN.Number * 10) + v_SubframeNumberT2;
    v_Duration := (v_MaxValue + v_MillisecondsT2 - v_MillisecondsT1) mod v_MaxValue;  // calculation considers wrap-around

    return v_Duration;

  }
After change

  function f_EUTRA_NB_SubFrameTimingDuration(SubFrameTiming_Type p_T1,

                                             SubFrameTiming_Type p_T2) return integer

  {

    var integer v_MillisecondsT1;

    var integer v_MillisecondsT2;

    var integer v_SubframeNumberT1 := 0;

    var integer v_SubframeNumberT2 := 0;

    var integer v_MaxValue := 1024*10240;   // SFN is 0..1023 in steps of 10ms

    var integer v_Duration;

    if (ischosen(p_T1.Subframe.Number) and

        ischosen(p_T2.Subframe.Number)) {

      v_SubframeNumberT1 := p_T1.Subframe.Number;

      v_SubframeNumberT2 := p_T2.Subframe.Number;

    }

    v_MillisecondsT1 := (p_T1.HSFN.Number*10240) + (p_T1.SFN.Number * 10) + v_SubframeNumberT1;
    v_MillisecondsT2 := (p_T2.HSFN.Number*10240) + (p_T2.SFN.Number * 10) + v_SubframeNumberT2;
    v_Duration := (v_MaxValue + v_MillisecondsT2 - v_MillisecondsT1) mod v_MaxValue;  // calculation considers wrap-around

    return v_Duration;

  }
2.2 Change 2 – Correction to altstep ‘a_AspFromTestcaseHandler_NBIOT’

	Altstep name
	a_AspFromTestcaseHandler_NBIOT()

	Reason for change
	Incorrect path for Dcch.

	Summary of change
	Incorrect path for Dcch.

	TTCN module
	NasEmu_Main_NBIOT.ttcn

	MCC160 Comment
	Accepted


Before change

  altstep a_AspFromTestcaseHandler_NBIOT(inout NasSecurity_Type p_NasSecurityByRef) runs on NASEMU_NBIOT_PTC

  {

...
[] TC_SRB.receive(car_NBIOT_SignallingAspFromTC) -> value v_AspFromTC {
...
      if (ischosen(v_ReqAspCommonPart.RoutingInfo.L2Test)) {

        if (ischosen(v_RrcPduToSYS.Dcch)) {

          v_L2SDU_MSG.Ccch := encvalue(v_RrcPduToSYS.Dcch);

        } else {

          v_L2SDU_MSG.Ccch := encvalue(v_RrcPduToSYS.Ccch);

        }

        L2TEST.send(v_L2SDU_MSG);

      } else {

        SYS_SRB.send(cas_NBIOT_PduToSYS(v_ReqAspCommonPart, v_RrcPduToSYS));

      }

    }

  }
After change

  altstep a_AspFromTestcaseHandler_NBIOT(inout NasSecurity_Type p_NasSecurityByRef) runs on NASEMU_NBIOT_PTC

  {

...

[] TC_SRB.receive(car_NBIOT_SignallingAspFromTC) -> value v_AspFromTC {

...

      if (ischosen(v_ReqAspCommonPart.RoutingInfo.L2Test)) {

        if (ischosen(v_RrcPduToSYS.Dcch)) {

          v_L2SDU_MSG.Dcch := encvalue(v_RrcPduToSYS.Dcch);

        } else {

          v_L2SDU_MSG.Ccch := encvalue(v_RrcPduToSYS.Ccch);

        }

        L2TEST.send(v_L2SDU_MSG);

      } else {

        SYS_SRB.send(cas_NBIOT_PduToSYS(v_ReqAspCommonPart, v_RrcPduToSYS));

      }

    }

  }

2.3 Change 3 – Correction to template ‘cs_UL_AM_RLC_NB_r13’

	Template name
	cs_UL_AM_RLC_NB_r13

	Reason for change
	Since maxRetxThreshold_r13 is set to t4 for RLC test case, pass the parameter p_MaxRetxThreshold_r13 in cs_UL_AM_RLC_NB_r13.

	Summary of change
	Since maxRetxThreshold_r13 is set to t4 for RLC test case, pass the parameter p_MaxRetxThreshold_r13 in cs_UL_AM_RLC_NB_r13.

	TTCN module
	NBIOT_RLC_AM_Templates.ttcn

	MCC160 Comment
	Accepted


Before change

 template (value) UL_AM_RLC_NB_r13 cs_UL_AM_RLC_NB_r13 (T_PollRetransmit_NB_r13 p_T_PollRetransmit_NB_r13) := {

    t_PollRetransmit_r13        := p_T_PollRetransmit_NB_r13,

    maxRetxThreshold_r13        := t8
  };
After change
 template (value) UL_AM_RLC_NB_r13 cs_UL_AM_RLC_NB_r13 (T_PollRetransmit_NB_r13 p_T_PollRetransmit_NB_r13,

                                                         UL_AM_RLC_NB_r13.maxRetxThreshold_r13 p_MaxRetxThreshold_r13 := t8) := {

    t_PollRetransmit_r13        := p_T_PollRetransmit_NB_r13,

    maxRetxThreshold_r13        := p_MaxRetxThreshold_r13
  };
2.4 Change 4 – Correction to function ‘f_NBIOT_Generate_RLC_SDUs’

	Function name
	f_NBIOT_Generate_RLC_SDUs()

	Reason for change
	Incorrect index for return value.

	Summary of change
	i =  p_StartIndex + p_Number after ‘for loop’ and the index of p_RLC_Rec.RLC_DataList is form p_StartIndex to p_StartIndex + p_Number - 1, so change i to i-1.

	TTCN module
	NBIOT_RLC_AM_Testcases.ttcn

	MCC160 Comment
	Accepted


Before change

  function f_NBIOT_Generate_RLC_SDUs(inout RLC_SS_State p_RLC_Rec,

                                     integer p_Number,

                                     integer p_NoOctets,

                                     integer p_StartIndex := 0,

                                     NBIOT_IDLEUPDATED_STATE_Type p_TestLoopMode := STATE2A_NB_TESTLOOP_ModeG)

    runs on NBIOT_PTC return integer

  {

    var integer i;

    var template (value) NAS_MSG_Request_Type v_NasPduFromTC;

    var octetstring v_EncodedL3Msg;

    for (i := p_StartIndex; i < (p_StartIndex + p_Number); i:= i + 1) {

      // 1. get random data

      p_RLC_Rec.TxDataSize := p_NoOctets;

      f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

      // 2. build Container

      p_RLC_Rec.RLC_UL_DataList[i] := p_RLC_Rec.RLC_Data; // store the random data in the RLC record

      v_NasPduFromTC := f_NBIOT_L2_CPMode_DL_NAS_Message(p_TestLoopMode, p_RLC_Rec.RLC_UL_DataList[i]);

      //  3. encode and cipher ESM DATA TRANSPORT

      v_EncodedL3Msg := f_NBIOT_L2_CPMode_EncodeL3Message(v_NasPduFromTC); // FFS to be made dependent on ESM Data Transport/ SMS TPDU

      p_RLC_Rec.RLC_DataList[i] := v_EncodedL3Msg;  // all list elements will have the same length

    }

    // calculate length in octets to do allow doing the segmentation correctly

    return (lengthof (p_RLC_Rec.RLC_DataList[i]));

  }
After change

  function f_NBIOT_Generate_RLC_SDUs(inout RLC_SS_State p_RLC_Rec,

                                     integer p_Number,

                                     integer p_NoOctets,

                                     integer p_StartIndex := 0,

                                     NBIOT_IDLEUPDATED_STATE_Type p_TestLoopMode := STATE2A_NB_TESTLOOP_ModeG)

    runs on NBIOT_PTC return integer

  {

    var integer i;

    var template (value) NAS_MSG_Request_Type v_NasPduFromTC;

    var octetstring v_EncodedL3Msg;

    for (i := p_StartIndex; i < (p_StartIndex + p_Number); i:= i + 1) {

      // 1. get random data

      p_RLC_Rec.TxDataSize := p_NoOctets;

      f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

      // 2. build Container

      p_RLC_Rec.RLC_UL_DataList[i] := p_RLC_Rec.RLC_Data; // store the random data in the RLC record

      v_NasPduFromTC := f_NBIOT_L2_CPMode_DL_NAS_Message(p_TestLoopMode, p_RLC_Rec.RLC_UL_DataList[i]);

      //  3. encode and cipher ESM DATA TRANSPORT

      v_EncodedL3Msg := f_NBIOT_L2_CPMode_EncodeL3Message(v_NasPduFromTC); // FFS to be made dependent on ESM Data Transport/ SMS TPDU

      p_RLC_Rec.RLC_DataList[i] := v_EncodedL3Msg;  // all list elements will have the same length

    }

    // calculate length in octets to do allow doing the segmentation correctly

    return (lengthof (p_RLC_Rec.RLC_DataList[i-1]));

  }
